Notes on the back story of this document:

The *Series 200* BASIC program **XLATOR** was created by me at the time I was professionally working with those machines back in the 80's. By then we had developed many engineering programs for *Series 80* machines *(HP-86/87)* but a decision was made to abandon *Series 80* and use *Series 200* machines instead.

Those machines *(HP 9816/26/36)* were essentially incompatible with *Series 80*, their BASIC was very different in many important aspects and the binaries were utterly incompatible too (*Capricorn* vs. *68000*) so our many programs would have to be rewritten from scratch. It was worth it despite the sizable cost because the new machines were at least 15x-30x faster than the older ones and we had to remain competitive.

Instead of rewriting our programs from scratch, which would have been extremely costly in terms of time (and thus money) and error-prone, I wrote first this automated conversion program, **XLATOR**, a 1,179-line *Series 200* BASIC program which ran on the *HP 9816/26/36*, would accept as input a *Series 80* BASIC program (which could use a number of binaries) and would produce the equivalent *Series 200* BASIC program as its output.

It wasn't perfect, of course, some fine-tuning had still to be done manually, but it took away 90-95% of the conversion job and completely avoided many of the myriad errors a human translator would commit when working on a long, complex program. It worked well and was a significant time/money-saver.

The source code is adequately commented (in Spanish) so you'll be able to understand most of it. Enjoy !

Valentin Albillo, 05-09-2021

XLATOR M32K (main memory, up to 32K)

```
100   ! ************************************************************************
120   ! *
140   ! *       CONVERTIDOR HP-87/HP-9816
160   ! *
180   ! ************************************************************************
200   OPTION BASE 1
220   COM /Salida/ @Aux,Printer,@Sub
240   COM /Long/ Mbytes,Sbytes
260   COM /Funciones/ Nombre$[80],Funci
280   DIM Binarias$(21)[9],A$[512],B$[512],Funcion$(34)[5],T$[512]
300   DIM Aux$[80],Aux1$[80],Aux2$[80]
320   INTEGER Start,Long,I,N,X,B
340   !   Aux
360 Funciones:    DATA ABS,ACS,ASN,ATN,ATN2,CEIL,CHR$,COS,COT,CSC,DTR,EXP
380               DATA FLOOR,FP,INT,IP,LEN,LGT,LOG,MAX,MIN,NUM,POS,RMD
400               DATA RTD,SEC,SGN,SIN,SQR,TAB,TAN,UPC$,VAL,VAL$
420               RESTORE Funciones
440               READ Funcion$(*)
460 Binarias:     DATA EPS,FP (,INF,MAX (,MIN (,RMD (,RTD (,SEC (,UPC$ (
480               DATA REV$ (,TRIM$ (,SAR$ (,RPT$ (,HGL$ (,LWC$ (
500               DATA ROT$ (,NOC$ (,INVI$ (,MSUS$ (,KEY$,TECLA$
520               RESTORE Binarias
540               READ Binarias$(*)
560               Nbinarias=21
580 Statements:   DATA "ASSIGN# ","ASSIGN %N","CHAIN ","LOAD "
600               DATA "CREATE ","CREATE BDAT "
620               DATA "DISP USING ","OUTPUT 1 USING ","DISP ","OUTPUT 1;"
640               DATA "OFF KEY#","OFF KEY","PRINT# ","OUTPUT %N"
660               DATA "READ# ","ENTER %N"
680               DATA "SHORT ","REAL "
700               DATA "FN END","FN end"," END"," STOP","FN end","FN END"
720               DATA "FLOOR (","INT (","IP (","INT (","HGL?$ (","FNHGL@$ ("
740               Nstatements=15
760   PRINTER IS 1
780   PRINT CHR$(12);
800   Printer=1
820   DIM Do$[40],Dc$[40]
840   LINPUT "Identificador disco original ? (:HP8290X,704,0)",Do$
860   IF Do$="" THEN Do$=":HP8290X,704,0"
880   LINPUT "Identificador disco traducc. ? (:HP8290X,704,1)",Dc$
900   IF Dc$="" THEN Dc$=":HP8290X,704,1"
920   MASS STORAGE IS Do$
940   Datos_file(N$,Start,Long)
960   MASS STORAGE IS Dc$
980   INPUT "Nombre del archivo a crear ?",New$
1000  INPUT "Identificador impresora ?",Printer
1020  INPUT "Listo tambien el proceso de traduccion ? (S/N)",Trad$
1040  ON ERROR GOTO Exis1
1060          PURGE New$
1080 Exis1: ON ERROR GOTO Exis2
1100            PURGE "W"&New$
1120 Exis2: ON ERROR GOTO Exis3
1140            PURGE "S"&New$
1160 Exis3: OFF ERROR
1180  CREATE ASCII "W"&New$,2*Long
1200  CREATE ASCII "S"&New$,Long
1220  ASSIGN @Aux TO "W"&New$
1240  ASSIGN @Sub TO "S"&New$
1260  PRINT CHR$(12);
```

```
1280  Rprinter=Printer
1300  IF Trad$[1,1]="S" OR Trad$[1,1]="s" THEN
1320     PRINTER IS Printer
1340     ELSE
1360     PRINTER IS 1
1380     Rprinter=Printer
1400     Printer=1
1420     END IF
1440  ALLOCATE INTEGER Source(Long*128)
1460  MASS STORAGE IS Do$
1480  PHYREAD Start,Source(*)
1500  MASS STORAGE IS Dc$
1520  A$=""
1540  !-------------------------------------------------------------------------------
1560  !    CICLO PRINCIPAL
1580  !-------------------------------------------------------------------------------
1600               Funci=0
1620               Mbytes=0
1640               Sbytes=0
1660               Out("10 GINIT")       ! Inicializar graficos
1680               Lastlin=10
1700               Overlap=0
1720     FOR Sector=0 TO Long-1
1740       FOR Posi=4 TO 256
1760         Byte=FNByte(Posi+256*Sector,Source(*))
1780         IF Byte=13 THEN
1800            ON ERROR GOTO Fin
1820               Newlin=VAL(A$)
1840            OFF ERROR
1860            IF Newlin>32766 THEN
1880               DISP "CUIDADO: Newlin=";Newlin;", es mayor que 32766 "
1900               Overlap=1
1920               BEEP
1940               WAIT 1
1960               END IF
1980            IF Newlin<=Lastlin THEN
2000               IF NOT Overlap THEN DISP "CUIDADO: Lastlin=";Lastlin;",Newlin
=";Newlin
2020               Overlap=1
2040               BEEP
2060               WAIT 1
2080               END IF
2100            GOSUB Convertir_linea
2120            Lastlin=Linea
2140            A$=""
2160            Posi=Posi+3
2180            GOTO Nextbyte
2200            END IF
2220         ON ERROR GOTO Fin
2240           A$=A$&CHR$(Byte)
2260         OFF ERROR
2280  Nextbyte: NEXT Posi
2300         NEXT Sector
2320  Fin:   OFF ERROR
2340         PRINT USING Rayas
2360         Out(VAL$(Lastlin+1)&" END",@Aux,"M")
2380         DEALLOCATE Source(*)
2400         ASSIGN @Aux TO *
2420         ASSIGN @Sub TO *
2440         BEEP
```

```
2460        OUTPUT 1;CHR$(12);
2480        OUTPUT 1;"TRADUCCION COMPLETADA: ";
2500           DISP "MBYTES=";Mbytes;", SBYTES=";Sbytes
2520           WAIT 1
2540        IF NOT Overlap THEN
2560           OUTPUT 1;"NO ha habido error. Tiene dos opciones :"
2580           OUTPUT 1;"   -si quiere renumerar y juntar en un solo archivo los
segmentos MAIN y SUB         pulse 'CONT' ."
2600           OUTPUT 1;"   -si no, tiene ya los dos segmentos en Wname y Sname,
y los puede añadir uno       tras otro con GET Wname y GET Sname,ultima linea"
2620           PAUSE
2640           ELSE
2660           OUTPUT 1;" ha habido error. RENUMERANDO ..."
2680           END IF
2700        Printer=Rprinter
2720        PRINTER IS Printer
2740        Conjuntar(New$,2,2)
2760        DISP "TRADUCCION FINALIZADA: hacer GET """&New$&""" para cargarlo"
2780        BEEP
2800        PAUSE
2820        STOP
2840 Rayas:IMAGE 80("-")
2860  !-----------------------------------------------------------------------
2880  ! TADUCCION DE CADA LINEA
2900  !-----------------------------------------------------------------------
2920 Convertir_linea:  PRINT USING Rayas
2940                   PRINT FNDelarameo$(A$)
2960                   PRINT
2980  !-----------------------------------------------------------------------
3000  ! TRATAMIENTOS PREVIOS : eliminar ! y las @ entre comillas
3020  !-----------------------------------------------------------------------
3040  A$=FNReplace$(A$,"@ REM","!")                ! cambia @ REM por !
3060  A$=FNReplace$(A$," REM","!")                 ! cambia REM por !
3080  A$=FNDelrem$(A$)                             ! elimina comentarios
3100  A$=FNReplace$(FNSar$(A$,"@","(","(","@")     ! elimina "@"
3120  !-----------------------------------------------------------------------
3140  ! TRATAMIENTO DE IF-THEN-ELSE
3160  !-----------------------------------------------------------------------
3180  A$=FNTrim$(A$)
3200  P=POS(A$,"IF ")
3220  IF P THEN
3240     P1=POS(A$[P+1],"ELSE ")
3260     IF P1 THEN
3280        Aux$="@ ELSE @"
3300        IF NOT POS(A$[P1+1],"@") THEN
3320           IF FNDir(A$,P+P1+5,LEN(A$)) THEN Aux$=Aux$&"GOTO "
3340           END IF
3360        A$=A$[1,P+P1-1]&Aux$&A$[P+P1+5]
3380        END IF
3400     P2=POS(A$[P+1],"@")
3420     IF P1 OR P2 THEN
3440              P3=POS(A$,"THEN ")
3460              Aux$=""
3480              IF P1 THEN
3500                 IF FNDir(A$,P3+5,P+P1-1) THEN Aux$=Aux$&"GOTO "
3520                 END IF
3540              A$=A$[1,P3+4]&"@"&Aux$&A$[P3+5]&"@END IF"
3560              END IF
3580     END IF
3600 !-----------------------------------------------------------------------
```

```
3620 !   CAMBIO DE NOMBRE DE LAS VARIABLES MATRICIALES
3640 !-------------------------------------------------------------------------------
3660         A$=FNCambmat$(A$)
3680 !-------------------------------------------------------------------------------
3700 !   CONVERSION DE FUNCIONES BINARIAS
3720 !-------------------------------------------------------------------------------
3740    FOR I=1 TO Nbinarias
3760        A$=FNReplace$(A$,Binarias$(I),"FN"&Binarias$(I))
3780    NEXT I
3800 !-------------------------------------------------------------------------------
3820 !   CONVERSION DE STATEMENTS Y FUNCIONES DE NOMBRE DIFERENTE
3840 !-------------------------------------------------------------------------------
3860    RESTORE Statements
3880    FOR I=1 TO Nstatements
3900        READ Aux1$,Aux2$
3920        A$=FNReplace$(A$,Aux1$,Aux2$)
3940    NEXT I
3960 !-------------------------------------------------------------------------------
3980 !   PARTICION DE LAS MULTILINEAS Y PROCESAMIENTO DE CADA LINEA
4000 !-------------------------------------------------------------------------------
4020          Linea=VAL(A$)
4040          LOOP
4060              P=POS(A$,"@")
4080              EXIT IF NOT P
4100              Process_line(A$[1,P-1],Linea)          ! Procesar linea
4120              Linea=Linea+1
4140              A$=VAL$(Linea)&" "&FNTrim$(A$[P+1])
4160          END LOOP
4180          Process_line(A$,Linea)                     ! Procesar linea
4200          RETURN
4220 END
4240 !-------------------------------------------------------------------------------
4260 ! PROCESAMIENTO DE CADA LINEA (CASOS RESTANTES)
4280 !-------------------------------------------------------------------------------
4300 SUB Process_line(T$,Linea)
4320     COM /Salida/ @Aux,Printer,@Sub
4340     COM /Funciones/ Nombre$,Funci
4360     DIM A$[512]
4380     A$=T$
4400     A$=FNLabel$(A$,Linea,Onlylab) ! Parte las lineas con etiqueta
4420        IF Onlylab THEN SUBEXIT    ! si solo etiqueta, no mas proceso
4440     A$=FNOnkey$(A$)               ! Tratamiento de los ON KEY#
4460     A$=FNReplace$(A$,"%","@")        ! Arreglo de los ASSIGN @, etc
4480     A$=FNConvstat$(A$)            ! Conversion Statement>Subprograma
4500     A$=FNReplace$(A$,"#","<>")    ! Conversion de # en <>
4520     A$=FNReplace$(A$,"\"," DIV ") ! Conversion de division entera
4540     A$=FNWait$(A$)                ! Tratamiento de WAIT
4560     A$=FNMatzer$(A$)              ! Conversion de MAT A=ZER
4580     Funciones(A$,Sent)            ! Separacion de DEF FN
4600     Graficos(A$,Multi,Sent)       ! Tratamiento de graficos
4620     Teclado(A$,Multi)             ! Trata funciones de bloqueo del KBD
4640     Multilet(A$,Multi)            ! Particion de asignaciones multiples
4660     Linea=Linea+Multi
4680     IF NOT Multi AND NOT Sent THEN CALL Out(A$)
4700 SUBEND
4720 !-------------------------------------------------------------------------------
4740 ! FNDIR=1 si T$[U,V] contiene una address o variable, y 0 si no
4760 !-------------------------------------------------------------------------------
4780 DEF FNDir(T$,U,V)
4800 Excepciones:  DATA BEEP,RETURN,ALPHA,ALPHALL,GRAPH,GRAPHICS,GRAPHALL
```

```
4820                DATA CLEAR,DEG,RAD,GRAD,DISP,PRINT,END,FRAME,GCLEAR
4840                DATA AXES,GRID,INITIALIZE,PAUSE,PENUP,RANDOMIZE
4860                DATA REM,RESTORE,SETGU,SETUU,STOP,UNCLIP
4880                Nexcepcion=28
4900                RESTORE Excepciones
4920        DIM A$[512]
4940        A$=T$
4960        X1=U
4980        X2=V
5000        WHILE A$[X1;1]=" "
5020            X1=X1+1
5040        END WHILE
5060        WHILE A$[X2;1]=" "
5080            X2=X2-1
5100        END WHILE
5120        FOR I=1 TO Nexcepcion
5140            READ Aux$
5160            IF A$[X1,X2]=Aux$ THEN RETURN 0
5180        NEXT I
5200        FOR I=X1 TO X2
5220            SELECT A$[I;1]
5240              CASE "A" TO "Z","a" TO "z","0" TO "9","_"
5260              CASE ELSE
5280                   RETURN 0
5300              END SELECT
5320        NEXT I
5340        RETURN 1
5360 FNEND
5380 !-------------------------------------------------------------------------------
5400 ! FNTrim$(T$) vuelve con T$ sin espacios iniciales ni finales
5420 !-------------------------------------------------------------------------------
5440 DEF FNTrim$(T$)
5460      DIM X$[512]
5480      X$=T$
5500      INTEGER I,J
5520      I=0
5540 F:   I=I+1
5560      IF I>LEN(X$) THEN RETURN ""
5580      IF X$[I,I]=" " THEN F
5600      J=LEN(X$)+1
5620 E:   J=J-1
5640      IF X$[J,J]=" " THEN E
5660      RETURN X$[I,J]
5680 FNEND
5700 !-------------------------------------------------------------------------------
5720 ! FNCambmat cambia el nombre de las matrices: de A(*) a Amat(*)
5740 !-------------------------------------------------------------------------------
5760 DEF FNCambmat$(T$)
5780      DIM A$[512]
5800      A$=T$
5820      A$=FNQuitanum$(A$,Linea$)
5840      IF A$[1,6]="IMAGE " THEN RETURN T$
5860      ALLOCATE Aux$[32]
5880 Sigue:         P=POS(A$,"(")
5900      IF NOT P THEN Yasta
5920      A$[P,P]="¿"
5940      K=P-1
5960      B$=A$[K;1]
5980      Flag=0
6000      SELECT B$
```

```
6020         CASE "0" TO "9"
6040            Flag=1
6060         CASE "A" TO "Z","a" TO "z","_","$"
6080         CASE ELSE
6100              GOTO Sigue
6120      END SELECT
6140          Aux$=""
6160 Otro:    Aux$=B$&Aux$
6180          K=K-1
6200          IF K=0 THEN Ok
6220      B$=A$[K;1]
6240      SELECT B$
6260         CASE "0" TO "9"
6280              GOTO Otro
6300         CASE "A" TO "Z","a" TO "z","_"
6320              Flag=0
6340              GOTO Otro
6360      END SELECT
6380 Ok:  IF Aux$[1,2]="FN" OR Flag THEN Sigue
6400      K=P
6420      IF A$[P-1,P-1]="$" THEN K=P-1
6440      A$[K]="mat"&A$[K]
6460      GOTO Sigue
6480 Yasta:    RETURN Linea$&" "&FNSar$(A$,"¿","(")
6500 FNEND
6520 !-------------------------------------------------------------------------------
6540 ! FNSar$ reemplaza en T$ todas las ocurrencia de U$ por V$
6560 !-------------------------------------------------------------------------------
6580 DEF FNSar$(T$,U$,V$)
6600      DIM A$[512],B$[512],C$[512]
6620      A$=T$
6640      B$=U$
6660      C$=V$
6680      L=LEN(B$)
6700      L1=LEN(C$)
6720      P1=1
6740      LOOP
6760          P=POS(A$[P1],B$)
6780          EXIT IF NOT P
6800          P1=P1+P-1
6820          A$=A$[1,P1-1]&C$&A$[P1+L]
6840          P1=P1+L1
6860      END LOOP
6880      RETURN A$
6900 FNEND
6920 !-------------------------------------------------------------------------------
6940 ! FNConvstat convierte statements inexistentes en subprogramas
6960 !-------------------------------------------------------------------------------
6980 DEF FNConvstat$(T$)
7000        DIM A$[512]
7020        A$=FNTrim$(T$)
7040        DIM Aux$[80]
7060 Stat: DATA "AREAD ","AWRITE ","START CRT AT ","FAST LABEL ","SGCLEAR "
7080        DATA "READCHAR ","WRITECHAR ","PAGE ","BPLOT ","BREAD ","CURSOR "
7100        DATA "FXD ","WHERE ","XAXIS ","YAXIS "
7120        DATA "DUMP GRAPHICS","DUMP ALPHA","PLOTTER IS "
7140        DATA "LEE ","ESCRIBE ","ETIQUETA ","PINTADA ","TECHO "
7160        Nstat=23
7180        RESTORE Stat
7200        FOR I=1 TO Nstat
```

```
7220            READ Aux$
7240            IF POS(A$,Aux$) THEN RETURN FNSar$(A$,Aux$,"CALL "&FNSar$(FNTrim$
(Aux$)," ","_")&"(")&")"
7260        NEXT I
7280        Aux$="SET I/O "
7300        IF POS(A$,Aux$) THEN RETURN FNSar$(A$,Aux$,"CALL SET_IO(")&")"
7320        RETURN A$
7340  FNEND
7360  !-----------------------------------------------------------------------------
7380  ! PARTICION DE ASIGNACIONES MULTIPLES EN VARIAS LINEAS
7400  !-----------------------------------------------------------------------------
7420  SUB Multilet(T$,Multi)
7440      OPTION BASE 1
7460      COM /Salida/ @Aux,Printer,@Sub
7480      COM /Funciones/ Nombre$,Funci
7500      DIM A$[512],B$[32]
7520      A$=T$
7540      Linea=VAL(A$)
7560      P=POS(A$,",")
7580      Q=POS(A$,"=")
7600      IF NOT P OR NOT Q THEN SUBEXIT
7620      R=POS(A$,"IF ")
7640      S=POS(A$,"THEN ")
7660      IF NOT R OR NOT S THEN Simple_let
7680      P=POS(A$[S],",")
7700      Q=POS(A$[S],"=")
7720      IF NOT P OR NOT Q THEN SUBEXIT
7740      Iflag=1
7760      ALLOCATE Prev$[512]
7780      Prev$=A$[1,S+3]
7800      A$=FNTrim$(A$[S+4])
7820      GOTO If_let
7840 Simple_let:      Iflag=0
7860                  A$=FNQuitanum$(A$,Linea$)
7880 If_let:          IF A$[1,4]="LET " THEN A$=A$[5]
7900      Nparab=0
7920      Ncorab=0
7940      Nvar=0
7960      String=0
7980      I=1
8000      Beg=I
8020      IF NOT FNChar(A$,I) THEN SUBEXIT   ! debe ser : 10 ! A,B=2 o asi
8040      ALLOCATE Var$(80)[80],Valor$[256]
8060 Rep1:I=I+1
8080      IF FNChar(A$,I) THEN Rep1
8100      IF A$[I,I]="$" THEN
8120         String=1
8140         GOTO Rep1
8160         END IF
8180      IF A$[I,I]="(" THEN
8200         Nparab=1
8220         GOTO Cerrar
8240         END IF
8260 Sig4:   IF A$[I,I]="[" THEN
8280         Ncorab=1
8300         GOTO Cerrar
8320         END IF
8340 Sig3:   IF A$[I,I]="," THEN Coma
8360         IF Nvar=0 THEN SUBEXIT
8380         IF A$[I,I]<>"=" THEN SUBEXIT  ! no debe suceder nunca !?!
```

```
8400          Nvar=Nvar+1
8420          Var$(Nvar)=A$[Beg,I-1]
8440          Valor$=A$[I+1]
8460          Aux$="Ñ"
8480          IF String THEN Aux$="Ñ$"
8500          IF Iflag THEN
8520                   Out(Prev$)
8540                   Linea=Linea+1
8560                   END IF
8580          Out(VAL$(Linea)&" "&Aux$&"="&Valor$)
8600          FOR I=1 TO Nvar
8620              Out(VAL$(Linea+I)&" "&Var$(I)&"="&Aux$)
8640          NEXT I
8660          IF Iflag THEN
8680             Out(VAL$(Linea+I)&" END IF")
8700             END IF
8720          Multi=Nvar+2*Iflag
8740          SUBEXIT
8760 Coma:      Nvar=Nvar+1
8780            Var$(Nvar)=A$[Beg,I-1]
8800            Beg=I+1
8820            GOTO Rep1
8840 Cerrar:    !
8860 Cierra:    I=I+1
8880            GOSUB Skipquotes
8900            IF A$[I,I]="(" THEN
8920               Nparab=Nparab+1
8940               GOTO Cierra
8960               END IF
8980            IF A$[I,I]="[" THEN
9000               Ncorab=Ncorab+1
9020               GOTO Cierra
9040               END IF
9060            IF A$[I,I]=")" THEN
9080               Nparab=Nparab-1
9100               GOTO Cerrado
9120               END IF
9140            IF A$[I,I]="]" THEN
9160               Ncorab=Ncorab-1
9180               END IF
9200 Cerrado:      IF Nparab OR Ncorab THEN Cierra
9220               I=I+1
9240               GOTO Sig4
9260 Skipquotes:   IF A$[I,I]<>CHR$(34) THEN RETURN
9280 Qrep:         I=I+1
9300               IF A$[I,I]<>CHR$(34) THEN Qrep
9320               I=I+1
9340               RETURN
9360  SUBEND
9380  !-------------------------------------------------------------------------------
9400  !   FNQuitanum$ devuelve T$ sin el numero de linea, que queda en Linea$
9420  !-------------------------------------------------------------------------------
9440  DEF FNQuitanum$(T$,Linea$)
9460      DIM A$[512]
9480      A$=T$
9500      Linea$=VAL$(VAL(A$))
9520      P=POS(A$,Linea$)
9540      RETURN FNTrim$(A$[P+LEN(Linea$)])
9560  FNEND
9580  !-------------------------------------------------------------------------------
```

```
9600  ! FNCHAR(A$,I) vale 1 si A$[I,I] es A-Z,a-z,0-9, o _, sino vale 0
9620  !-------------------------------------------------------------------------------
9640  DEF FNChar(A$,I)
9660      SELECT A$[I,I]
9680             CASE "A" TO "Z","a" TO "z","0" TO "9","_"
9700                  RETURN 1
9720             CASE ELSE
9740                  RETURN 0
9760      END SELECT
9780  FNEND
9800  !-------------------------------------------------------------------------------
9820  ! MANDA LA LINEA A LA IMPRESORA, LA PANTALLA Y EL DISCO
9840  !-------------------------------------------------------------------------------
9860  SUB Out(A$,OPTIONAL @Dest,Dest$)
9880      COM /Salida/ @Aux,Printer,@Sub
9900      COM /Long/ Mbytes,Sbytes
9920      COM /Funciones/ Nombre$,Funci
9940      IF NPAR=2 THEN A=0/0 ! ************* TEMPORAL *******
9960      A$=FNSar$(A$,"{","@")
9980      DISP A$
10000     IF Printer=-1 THEN Nprint
10020     PRINT FNDelarameo$(A$)
10040     IF Printer<>1 THEN OUTPUT 1;FNDelarameo$(A$,1)
10060 Nprint:   L=LEN(A$)+2
10080           IF NPAR=3 THEN
10100              OUTPUT @Dest;A$
10120              IF Dest$="S" THEN Sbytes=Sbytes+L
10140              IF Dest$="M" THEN Mbytes=Mbytes+L
10160              SUBEXIT
10180              END IF
10200     IF Funci  THEN
10220              Sbytes=Sbytes+L
10240              OUTPUT @Sub;A$
10260              ELSE
10280              Mbytes=Mbytes+L
10300              OUTPUT @Aux;A$
10320              END IF
10340 SUBEND
10360 !-------------------------------------------------------------------------------
10380 ! FNReplace$(T$,U$,V$) devuelve T$ con los U$ cambiados por V$
10400 !                        EXCEPTO CUANDO APARECEN ENTRE COMILLAS
10420 !-------------------------------------------------------------------------------
10440 DEF FNReplace$(T$,U$,V$)
10460     DIM A$[512],B$[512],C$[512]
10480     A$=T$
10500     B$=U$
10520     C$=V$
10540     I=1
10560 Rep3:P=POS(A$[I],B$)
10580     IF NOT P THEN
10600              IF I=1 THEN RETURN A$
10620              RETURN FNSar$(A$,"¿"&B$[2],C$)
10640              END IF
10660     P=P+I-1
10680     Q=POS(A$[I],CHR$(34))
10700     IF NOT Q THEN
10720              A$[P]=FNSar$(A$[P],B$,C$)
10740              RETURN FNSar$(A$,"¿"&B$[2],C$)
10760              END IF
10780     Q=Q+I-1
```

```
10800 Rep:IF Q>P THEN
10820             A$[P,P]="¿"
10840             P=POS(A$[I],B$)
10860             IF NOT P THEN RETURN FNSar$(A$,"¿"&B$[2],C$)
10880             P=P+I-1
10900             GOTO Rep
10920             END IF
10940 Rep2:Q=Q+1
10960       IF A$[Q,Q]<>CHR$(34) THEN Rep2
10980       I=Q+1
11000       GOTO Rep3
11020 FNEND
11040 !-----------------------------------------------------------------------
11060 ! ALMACENA POR SEPARADO FUNCIONES DEFINIDAS EN EL PROGRAMA
11080 !-----------------------------------------------------------------------
11100 SUB Funciones(T$,Sent)
11120     COM /Salida/ @Aux,Printer,@Sub
11140     COM /Funciones/ Nombre$,Funci
11160     DIM A$[512]
11180     Sent=1
11200     A$=FNTrim$(T$)
11220     A$=FNQuitanum$(A$,Linea$)
11240     Linea=VAL(Linea$)
11260     IF A$[1,6]="DEF FN" THEN Header
11280     IF POS(A$,Nombre$) THEN Value
11300     IF A$[1,6]="FN END" THEN Final
11320     Sent=0
11340     SUBEXIT
11360 Value:      Aux$="ÑÑ"
11380             IF POS(Nombre$,"$") THEN Aux$="ÑÑ$"
11400             Out(Linea$&" "&FNReplace$(A$,Nombre$,Aux$),@Sub,"S")
11420             SUBEXIT
11440 Final:      Aux$="ÑÑ"
11460             IF POS(Nombre$,"$") THEN Aux$="ÑÑ$"
11480             Out(Linea$&" RETURN "&Aux$,@Sub,"S")
11500             Out(VAL$(Linea+1)&" FNEND",@Sub,"S")
11520             Funci=0
11540             SUBEXIT
11560 Header:     A$=FNDelcor$(A$)
11580             P=POS(A$,"=")
11600             IF P THEN
11620          !-----------------------------------------------
11640          ! UNA SOLA LINEA
11660          !-----------------------------------------------
11680                Out(Linea$&" "&A$[1,P-1],@Sub,"S")
11700                Out(VAL$(Linea+1)&" RETURN "&A$[P+1],@Sub,"S")
11720                Out(VAL$(Linea+2)&" FNEND",@Sub,"S")
11740                Funci=0
11760                SUBEXIT
11780                END IF
11800          !-----------------------------------------------
11820          !  FUNCIONES MULTILINEA
11840          !-----------------------------------------------
11860               P=POS(A$,"(")
11880               IF NOT P THEN
11900                        Nombre$=A$[5]
11920                        ELSE
11940                        Nombre$=A$[5,P-1]
11960                        END IF
11980               Out(Linea$&" "&A$,@Sub,"S")
```

```
12000               Funci=1
12020               SUBEXIT
12040   SUBEND
12060   !-----------------------------------------------------------------------
12080   ! FNDelcor$ quita los corchetes y su contenido en los parametros
12100   !           de una funcion: DEF FNA$(A$[80],B$)>DEF FNA$(A$,B$)
12120   !-----------------------------------------------------------------------
12140   DEF FNDelcor$(T$)
12160       DIM A$[512]
12180       A$=T$
12200       LOOP
12220         P=POS(A$,"[")
12240         EXIT IF NOT P
12260         Q=POS(A$,"]")
12280         A$=A$[1,P-1]&A$[Q+1]
12300       END LOOP
12320       RETURN A$
12340   FNEND
12360   !-----------------------------------------------------------------------
12380   ! FNWait(T$) cambia 10 WAIT 200+7 en 10 WAIT (200+7)/1000
12400   !-----------------------------------------------------------------------
12420   DEF FNWait$(T$)
12440       DIM A$[512],B$[512]
12460       P=POS(T$,"WAIT ")
12480       IF NOT P THEN RETURN T$
12500       A$=FNTrim$(T$)
12520       B$=FNReplace$(A$,"WAIT ","WAIT (")
12540       IF A$<>B$ THEN RETURN B$&")/1E3"
12560       RETURN A$
12580   FNEND
12600   !-----------------------------------------------------------------------
12620   ! TRATAMIENTO DE LAS INSTRUCCIONES GRAFICAS Y DE PANTALLA
12640   !-----------------------------------------------------------------------
12660   SUB Graficos(T$,Multi,Sent)
12680       DIM A$[512]
12700       A$=FNTrim$(T$)
12720       Linea=VAL(A$)
12740       Multi=0
12760       IF POS(A$," CLEAR") THEN
12780          Out(FNReplace$(A$," CLEAR"," CALL Aclear"))
12800          Sent=1
12820          SUBEXIT
12840          END IF
12860       IF POS(A$," GSTORE ") THEN
12880          Out(FNReplace$(A$," GSTORE "," CALL GSTORE_(")&")")
12900          Sent=1
12920          SUBEXIT
12940          END IF
12960       IF POS(A$," GLOAD ") THEN
12980          Out(FNReplace$(A$," GLOAD "," CALL GLOAD_(")&")")
13000          Sent=1
13020          SUBEXIT
13040          END IF
13060       IF POS(A$," ALPHA") OR POS(A$," GRAPH") THEN
13080         IF POS(A$," ALPHALL") THEN A$=FNReplace$(A$," ALPHALL"," ALPHA")
13100         IF POS(A$," GRAPHALL") THEN A$=FNReplace$(A$," GRAPHALL"," GRAPH")
13120         IF POS(A$," GRAPHICS") THEN A$=FNReplace$(A$," GRAPHICS"," GRAPH")
13140         Aflag=0
13160         IF POS(A$," ALPHA") THEN Aflag=1
13180             Iflag=0
```

```
13200             IF POS(A$,"IF ") AND POS(A$,"THEN ") THEN Iflag=1
13220             A$=FNQuitanum$(A$,Linea$)
13240             IF Iflag THEN
13260                      P=POS(A$,"THEN ")
13280                      Out(Linea$&" "&A$[1,P+3])
13300                      A$=A$[P+5]
13320                      Linea=Linea+1
13340                      END IF
13360             IF Aflag THEN
13380                Out(VAL$(Linea)&" "&FNReplace$(A$,"ALPHA","ALPHA ON"))
13400                Out(VAL$(Linea+1)&" GRAPHICS OFF")
13420                ELSE
13440                Out(VAL$(Linea)&" "&FNReplace$(A$,"GRAPH","GRAPHICS ON"))
13460                Out(VAL$(Linea+1)&" ALPHA OFF")
13480                END IF
13500             IF Iflag THEN CALL Out(VAL$(Linea+2)&" END IF")
13520             Multi=1+2*Iflag
13540             SUBEXIT
13560          END IF
13580  SUBEND
13600  !-------------------------------------------------------------------------------
13620  ! FNMatzer(T$) convierte 10 MAT PEPE=ZER en 10 CALL MATZER(PEPE(*))
13640  !-------------------------------------------------------------------------------
13660  DEF FNMatzer$(T$)
13680      DIM A$[512]
13700      A$=T$
13720      P=POS(A$," MAT ")
13740      Q=POS(A$,"=ZER")
13760      IF NOT P OR NOT Q OR Q<P OR Q-P>37 THEN RETURN A$
13780      A$=A$[1,Q-1]
13800      RETURN FNReplace$(A$," MAT "," CALL MATZER(")&"mat(*))"
13820  FNEND
13840  !-------------------------------------------------------------------------------
13860  ! FNLabel$ separa y trata las lineas etiquetadas, y las manda al disco
13880  !-------------------------------------------------------------------------------
13900  DEF FNLabel$(T$,Linea,Onlylab)
13920      DIM A$[512]
13940      A$=T$
13960      Onlylab=0
13980      IF NOT POS(A$,":") THEN RETURN T$
14000      A$=FNQuitanum$(A$,Linea$)
14020      I=1
14040      REPEAT
14060        I=I+1
14080      UNTIL NOT FNChar(A$,I)
14100      IF A$[I,I]<>":" THEN RETURN T$
14120      Out(Linea$&" "&A$[1,I]&"!")
14140      L=LEN(A$)
14160      REPEAT
14180        I=I+1
14200        IF I>L THEN
14220               Onlylab=1
14240               RETURN ""
14260               END IF
14280      UNTIL A$[I,I]<>" "
14300      Linea=Linea+1
14320      RETURN VAL$(Linea)&" "&A$[I]
14340  FNEND
14360  !-------------------------------------------------------------------------------
14380  ! FNOnkey transforma ON KEY#3,"PEPE" GOTO 200 en su equivalente
```

```
14400  !-------------------------------------------------------------------------------
14420  DEF FNOnkey$(T$)
14440      DIM A$[512],B$[512]
14460      A$=T$
14480      P=POS(A$,"ON KEY# ")
14500      IF NOT P THEN RETURN T$
14520      B$=A$[P]
14540      B$=FNReplace$(FNReplace$(B$,"#",""),",","," LABEL ")
14560      RETURN A$[1,P-1]&B$
14580  FNEND
14600  !-------------------------------------------------------------------------------
14620  ! FNDelrem$ elimina los comentarios: suprime lo que sigue a !
14640  !-------------------------------------------------------------------------------
14660  DEF FNDelrem$(T$)
14680      DIM A$[512],B$[512]
14700      A$=T$
14720      P=POS(A$,"!")
14740      IF NOT P THEN RETURN T$
14760      B$=FNQuitanum$(A$,Linea$)
14780      IF B$[1,1]="!" THEN RETURN A$[1,P]
14800      Q=POS(A$,CHR$(34))
14820      IF NOT Q OR Q>P THEN RETURN A$[1,P-1]
14840      A$=FNReplace$(FNSar$(A$,"!","{"),"{","!")
14860      P=POS(A$,"!")
14880      IF NOT P THEN RETURN T$
14900      RETURN FNSar$(A$[1,P-1],"{","!")
14920  FNEND
14940 !-------------------------------------------------------------------------------
14960 ! FNByte proporciona el valor decimal del byte N del array entero A(*)
14980 !-------------------------------------------------------------------------------
15000 DEF FNByte(INTEGER N,A(*))
15020     K=N+1
15040     B=A(K DIV 2)
15060     IF B<0 THEN B=B+65536
15080     IF K MOD 2 THEN RETURN B MOD 256
15100     RETURN B DIV 256
15120 FNEND
15140 !-------------------------------------------------------------------------------
15160 !  CAT e INFO son utilidades para extraer informacion del catalogo
15180 !-------------------------------------------------------------------------------
15200 SUB Info(File$,INTEGER Dstart,Dleng,Strt,Stp,Lrecs,L_rec_l,Ftype,OPTIONAL
A$(*),INTEGER Files)
15220 INTEGER Address,Length,L_record,I,J,Type,Numfiles
15240 DIM Filename$[10]
15260 Type=1
15280 Address=2
15300 Length=3
15320 L_record=7
15340 Strt=-1
15360 Filename$=FNTrim$(File$)
15380 IF NPAR>8 THEN
15400  ALLOCATE B$(Dleng*64-1)[18]
15420  FOR I=0 TO Dleng*64-1
15440    B$(I)=A$(I)
15460  NEXT I
15480  Numfiles=Files
15500 ELSE
15520  ALLOCATE INTEGER Sector0(127)
15540  PHYREAD 0,Sector0(*)
15560  Dstart=Sector0(5)
```

```
15580  IF Dstart<0 THEN Dstart=Dstart+65536
15600  Dstart=Sector0(4)*65536+Dstart
15620  Dleng=Sector0(9)
15640  IF Dleng<0 THEN Dleng=Dleng+65536
15660  Dleng=Sector0(8)*65536+Dleng
15680  DEALLOCATE Sector0(*)
15700  ALLOCATE B$(Dleng*64-1)[18]
15720! LOADSUB ALL FROM "CAT"
15740  CALL Cat(B$(*),Dstart,Dleng,Numfiles)
15760! DELSUB Cat TO END
15780 END IF
15800 Search:                    ! *
15820 FOR J=0 TO (Numfiles-1)*8 STEP 8
15840  IF (FNTrim$(B$(J))=Filename$) THEN
15860    IF B$(J+Type)="ASCII" THEN Ftype=1
15880    IF B$(J+Type)="BIN" THEN Ftype=-5775
15900    IF B$(J+Type)="BDAT" THEN Ftype=-5791
15920    IF B$(J+Type)="PROG" THEN Ftype=-5808
15940    IF B$(J+Type)="SYSTM" THEN Ftype=-5822
15960    Strt=VAL(B$(J+Address))
15980    Stp=Strt+VAL(B$(J+Length))-1
16000    Lrecs=VAL(B$(J+Length))
16020    L_rec_l=VAL(B$(J+L_record))
16040    IF Ftype=-5791 THEN
16060      Strt=Strt-1
16080      IF L_rec_l=0 THEN L_rec_l=256
16100      IF L_rec_l<>1 THEN
16120        IF L_rec_l MOD 2 THEN L_rec_l=L_rec_l+1
16140      END IF
16160      Needed_recs=Lrecs*1.0%L_rec_l
16180      IF Needed_recs MOD 256 THEN
16200        Needed_recs=INT(Needed_recs/256+1)+1
16220      ELSE
16240        Needed_recs=Needed_recs/256+1
16260      END IF
16280      Stp=Strt+Needed_recs-1
16300    END IF
16320    SUBEXIT
16340  END IF
16360 NEXT J
16380 SUBEND!


16400 SUB Cat(A$(*),INTEGER Dstart,Dleng,Num_files)
16420 INTEGER Sector,Entry,Masklo,Maskhi,I
16440 ALLOCATE INTEGER A(Dleng-1,127)
16460 Num_files=0
16480 Masklo=255
16500 Maskhi=-256
16520 ON ERROR GOSUB Error
16540 PHYREAD Dstart,A(*)
16560 Entry_ptr=-1
16580 FOR Sector=0 TO Dleng-1
16600  FOR Entry=0 TO 7
16620 Filetype:     !
16640    IF A(Sector,Entry*16+5)=-1 THEN Eod
16660    IF A(Sector,Entry*16+5)=0 THEN Next_entry
16680    Num_files=Num_files+1
16700    Entry_ptr=Entry_ptr+1
16720    IF A(Sector,Entry*16+5)=1 THEN A$(Entry_ptr*8+1)="ASCII"
```

```
16740     IF A(Sector,Entry*16+5)=-5775 THEN A$(Entry_ptr*8+1)="BIN"
16760     IF A(Sector,Entry*16+5)=-5791 THEN A$(Entry_ptr*8+1)="BDAT"
16780     IF A(Sector,Entry*16+5)=-5808 THEN A$(Entry_ptr*8+1)="PROG"
16800     IF A(Sector,Entry*16+5)=-5822 THEN A$(Entry_ptr*8+1)="SYSTM"
16820 Filename:     !
16840     FOR I=1 TO 5
16860       A$(Entry_ptr*8)[I*2-1]=CHR$(SHIFT(BINAND(A(Sector,Entry*16+(I-1)),Mas
khi),8))
16880       A$(Entry_ptr*8)[I*2]=CHR$(BINAND(A(Sector,Entry*16+(I-1)),Masklo))
16900     NEXT I
16920 Start_address:!
16940     IF A(Sector,Entry*16+5)=-5791 THEN
16960       A$(Entry_ptr*8+2)=VAL$(A(Sector,Entry*16+7)+1)
16980     ELSE
17000       A$(Entry_ptr*8+2)=VAL$(A(Sector,Entry*16+7))
17020     END IF
17040 Filelength:!
17060     IF A(Sector,Entry*16+5)=-5791 THEN
17080       ALLOCATE INTEGER B(127)
17100       PHYREAD A(Sector,Entry*16+7),B(*)
17120       IF B(4)<>0 OR B(5)<0 THEN
17140         A$(Entry_ptr*8+3)="****"
17160       ELSE
17180         A$(Entry_ptr*8+3)=VAL$(B(5))
17200       END IF
17220       DEALLOCATE B(*)
17240     ELSE
17260       A$(Entry_ptr*8+3)=VAL$(A(Sector,Entry*16+9))
17280     END IF
17300 Creation_time:!
17320     A$(Entry_ptr*8+4)[1,4]=VAL$(A(Sector,Entry*16+10))
17340     A$(Entry_ptr*8+4)[5,8]=VAL$(A(Sector,Entry*16+11))
17360     A$(Entry_ptr*8+4)[9,12]=VAL$(A(Sector,Entry*16+12))
17380 Vol_info:!
17400     A$(Entry_ptr*8+5)=VAL$(A(Sector,Entry*16+13))
17420 Protect_code:!
17440     IF A(Sector,Entry*16+5)<>1 AND VAL$(A(Sector,Entry*16+14))<>"8224" THEN

17460       A$(Entry_ptr*8+6)="*"
17480     ELSE
17500       A$(Entry_ptr*8+6)=" "
17520     END IF
17540 Log_rec_size:!
17560     IF A(Sector,Entry*16+5)=-5791 THEN
17580       IF A(Sector,Entry*16+15)=0 THEN
17600         A$(Entry_ptr*8+7)="1"
17620       ELSE
17640         A$(Entry_ptr*8+7)=VAL$(A(Sector,Entry*16+15)*2)
17660       END IF
17680     ELSE
17700       A$(Entry_ptr*8+7)="256"
17720     END IF
17740 Next_entry:     !
17760   NEXT Entry
17780 NEXT Sector
17800 Eod:!
17820 OFF ERROR
17840 SUBEXIT
17860 Error:!
17880 BEEP
```

```
17900 PRINT FNLin$(8);"Error ";ERRN
17920 PRINT FNLin$(2);"Remedy error condition and press CONTINUE"
17940 PAUSE
17960 RETURN
17980 SUBEND!


18000 DEF FNLin$(INTEGER X)
18020 INTEGER I
18040 IF X=0 THEN RETURN CHR$(13)
18060 Eol$=CHR$(13)&CHR$(10)
18080 IF X<0 THEN Eol$=CHR$(10)
18100 ALLOCATE R$[X*LEN(Eol$)]
18120 R$=""
18140 FOR I=1 TO X
18160  R$=R$&Eol$
18180 NEXT I
18200 RETURN R$
18220 FNEND!


18240!-------------------------------------------------------------------------------
18260! DATOS_FILE vuelve con la Start y la Long de un archivo dado
18280!-------------------------------------------------------------------------------
18300 SUB Datos_file(N$,INTEGER Start,Long)
18320    OPTION BASE 0
18340    INTEGER Nfiles,Final,Lwords,Tipo
18360    CAT
18380 Nombre:    INPUT "Nombre del archivo a convertir ?",N$
18400            Info(N$,2,14,Start,Final,Long,Lwords,Tipo)
18420    IF Start=-1 THEN
18440       BEEP
18460       DISP "El archivo "&N$&" no esta en este disco !!! . Repita ..."
18480       WAIT 1
18500       GOTO Nombre
18520       END IF
18540 SUBEND
18560 !-------------------------------------------------------------------------------
18580 ! Teclado trata TAKE KEYBOARD y RELEASE KEYBOARD
18600 !-------------------------------------------------------------------------------
18620 SUB Teclado(T$,Multi)
18640     DIM A$[512],Aux1$[30],Aux2$[30]
18660     A$=T$
18680     Iflag=0
18700 Take:    DATA EMBARAZA TECLAS,TAKE KEYBOARD,JODE TECLADO,JODETE TECLADO
18720          Ntake=4
18740 Release:DATA ABORTA TECLAS,RELEASE KEYBOARD,RELAX TECLADO,RELAJATE TECLADO
18760          RESTORE Take
18780          FOR I=1 TO Ntake
18800              READ Aux$
18820              IF POS(A$,Aux$) THEN
18840                 Aux1$="SUSPEND INTERACTIVE"
18860                 Aux2$="ON KBD ALL,15 CALL Dummy"
18880                 GOTO Vale
18900                 END IF
18920          NEXT I
18940          RESTORE Release
18960          FOR I=1 TO Ntake
18980              READ Aux$
19000              IF POS(A$,Aux$) THEN
```

```
19020                   Aux1$="RESUME INTERACTIVE"
19040                   Aux2$="OFF KBD"
19060                   GOTO Vale
19080                   END IF
19100           NEXT I
19120           SUBEXIT
19140 Vale:     A$=FNQuitanum$(A$,Linea$)
19160           Linea=VAL(Linea$)
19180           IF A$[1,3]="IF " THEN Iflag=1
19200           IF Iflag THEN
19220                     P=POS(A$,"THEN ")
19240                     Out(Linea$&" "&A$[1,P+3])
19260                     A$=A$[P+5]
19280                     Linea=Linea+1
19300                     END IF
19320     Out(VAL$(Linea)&" "&FNSar$(A$,Aux$,Aux1$))
19340     Out(VAL$(Linea+1)&" "&Aux2$)
19360     IF Iflag THEN CALL Out(VAL$(Linea+2)&" END IF")
19380     Multi=1+2*Iflag
19400 SUBEND
19420 !-------------------------------------------------------------------------------
19440 ! FNDelarameo$ evita que la impresora opcional escriba en arameo
19460 !-------------------------------------------------------------------------------
19480 DEF FNDelarameo$(T$,OPTIONAL P)
19500       DIM A$[512],B$[512]
19520       A$=T$
19540       B$=""
19560       Flag=0
19580       FOR I=1 TO LEN(A$)
19600           A=NUM(A$[I,I])
19620           IF A<32 THEN
19640                     IF NPAR=1 THEN Salta
19660                     Flag=1
19680                     END IF
19700           B$=B$&CHR$(A)
19720 Salta:    NEXT I
19740           CONTROL 1,4;Flag
19760           RETURN B$
19780 FNEND
19800 !-------------------------------------------------------------------------------
19820 ! RENUMERAR renumera @A , utilizando @F como auxiliar
19840 !-------------------------------------------------------------------------------
19860 SUB Renumerar(File$,Name$,Flin,Inc)
19880       OPTION BASE 1
19900       COM /Salida/ @Aux,Printer,@Sub
19920       COM /Funciones/ Nombre$,Funci
19940       DIM Lineas(1000),A$[512],B$[512]
19960       Rprinter=Printer
19980       Printer=-1
20000       ASSIGN @A TO Name$
20020       Poin=1
20040       ON END @A GOTO Pas1
20060 Ent:ENTER @A;A$
20080       Append(A$,File$,Poin)
20100       GOTO Ent
20120 Pas1:OFF END @A
20140        ASSIGN @A TO *
20160        File$[Poin;1]=CHR$(0)   ! Fin
20180        ALLOCATE Auxil$[1.1*LEN(File$)]
20200       !-----------------------------------------
```

```
20220      ! PRIMERA PASADA: Cambiar direcciones
20240      !-----------------------------------
20260      OUTPUT 1;"    --PRIMERA PASADA"
20280      N=1
20300      Poin1=1
20320      Poin2=1
20340 Otralin:     IF NOT FNExtract(A$,File$,Poin1) THEN Pas2
20360      Aux$=" GOTO "
20380      IF POS(A$,Aux$) THEN
20400         IF POS(A$," ON ") THEN Es_on
20420         GOTO Es_go
20440         END IF
20460      Aux$=" GOSUB "
20480      IF POS(A$,Aux$) THEN
20500         IF POS(A$," ON ") THEN Es_on
20520         GOTO Es_go
20540         END IF
20560      Aux$=" THEN "
20580      P=POS(A$,Aux$)
20600      IF POS(A$," IF ") AND P AND LEN(A$)>P+5 THEN Es_go
20620      IF POS(A$," END") THEN
20640         B$=FNQuitanum$(A$,Linea$)
20660         IF B$="END" THEN Pasada2
20680         END IF
20700 Ou: Append(A$,Auxil$,Poin2)
20720      GOTO Otralin
20740 Es_go:    P=POS(A$,Aux$)
20760           L=LEN(Aux$)
20780 On_esp:   IF NOT FNDirnum(A$[P+L]) THEN Ou
20800           Linea=VAL(A$[P+L])
20820           GOSUB Storeli
20840           A$[P+L]="Ñ"&VAL$(I)
20860           GOTO Ou
20880 Es_on:    P=POS(A$,Aux$)
20900           L=LEN(Aux$)
20920           K=P+L
20940 Sigdir:   Q=POS(A$[P+L],",")
20960           IF NOT Q THEN
20980              A$[K]=FNSar$(A$[K],"{",",")
21000              GOTO On_esp
21020              END IF
21040           A$[P+L+Q-1;1]="{"
21060           IF FNDirnum(A$[P+L;Q-1]) THEN
21080              Linea=VAL(A$[P+L;Q-1])
21100              GOSUB Storeli
21120              Nlin$="Ñ"&VAL$(I)
21140              A$=A$[1,P+L-1]&Nlin$&A$[P+L+Q-1]
21160              P=P+LEN(Nlin$)+1
21180              GOTO Sigdir
21200              END IF
21220           P=P+Q
21240           GOTO Sigdir
21260 Storeli:FOR I=1 TO N-1
21280              IF Lineas(I)=Linea THEN RETURN
21300         NEXT I
21320         N=N+1
21340         Lineas(I)=Linea
21360         RETURN
21380 Pasada2:  Append(A$,Auxil$,Poin2)
21400 Pas2:     Auxil$[Poin2;1]=CHR$(0)
```

```
21420           Nlabels=N-1
21440         !-------------------------------------------------------------
21460         ! SEGUNDA PASADA: renumera las lineas y añade etiquetas
21480         !-------------------------------------------------------------
21500         OUTPUT 1;"   --SEGUNDA PASADA"
21520  Poin1=1
21540  Poin2=1
21560  Nlinea=Flin
21580  Eflag=0
21600 Main_loop: IF NOT FNExtract(A$,Auxil$,Poin2) THEN Sacabo
21620            Nlin$=VAL$(Nlinea)
21640            A$=FNQuitanum$(A$,Linea$)
21660            Linea=VAL(Linea$)
21680            IF A$="END" THEN Eflag=1
21700            Lflag=0
21720            IF POS(A$,":") THEN Lflag=1
21740            FOR I=1 TO Nlabels
21760                IF Lineas(I)=Linea THEN
21780                   Aux$=" Ñ"&VAL$(I)&": "
21800                   IF Lflag THEN
21820                      J=1
21840                      REPEAT
21860                        J=J+1
21880                      UNTIL NOT FNChar(A$,J)
21900                      IF A$[J,J]=":" THEN
21920                         Append(VAL$(Nlinea-1)&Aux$&"!",File$,Poin1)
21940                         Aux$=" "
21960                         END IF
21980                   END IF
22000                   GOTO Outlin
22020                END IF
22040          NEXT I
22060          Aux$=" "
22080 Outlin:  Append(Nlin$&Aux$&A$,File$,Poin1)
22100          Nlinea=Nlinea+Inc
22120          IF NOT Eflag THEN Main_loop
22140 Sacabo:  File$[Poin1;1]=CHR$(0)
22160          Flin=Nlinea
22180 SUBEND
22200!-------------------------------------------------------------------------
22220! FNDirnum vuelve con 1 si T$ es 1-99999, y con 0 en caso contrario
22240!-------------------------------------------------------------------------
22260 DEF FNDirnum(T$)
22280    DIM A$[512]
22300    A$=FNTrim$(T$)
22320    IF LEN(A$)<1 OR LEN(A$)>5 THEN RETURN 0
22340    FOR I=1 TO LEN(A$)
22360        IF A$[I,I]<"0" OR A$[I,I]>"9" THEN RETURN 0
22380    NEXT I
22400    RETURN 1
22420 FNEND
22440!-------------------------------------------------------------------------
22460! CONJUNTAR renumera y junta en un solo archivo el MAIN y las DEF FN
22480!-------------------------------------------------------------------------
22500 SUB Conjuntar(New$,Flin,Inc)
22520     COM /Salida/ @Aux,Printer,@Sub
22540     COM /Long/ Mbytes,Sbytes
22560     COM /Funciones/ Nombre$,Funci
22580     DIM A$[512]
22600     ALLOCATE Main$[1.1*Mbytes]
```

```
22620      IF Sbytes THEN ALLOCATE Sub$[1.1*Sbytes]
22640      Rprinter=Printer
22660      OUTPUT 1;"** RENUMERANDO MAIN"
22680      Renumerar(Main$,"W"&New$,Flin,Inc)
22700      L1=LEN(Main$)
22720      L2=0
22740      IF Sbytes THEN
22760         OUTPUT 1;"** RENUMERANDO FUNCIONES"
22780         Renumerar(Sub$,"S"&New$,Flin,Inc)
22800         L2=LEN(Sub$)
22820         END IF
22840      DISP
22860      CREATE ASCII New$,INT(1.1*(L1+L2)/256)+1
22880      ASSIGN @Final TO New$
22900      Printer=Rprinter
22920      PRINTER IS Printer
22940      PRINT "**** TRADUCCION COMPLETADA ****"
22960      PRINT "-----------------------------------"
22980      PRINT USING "5/"
23000      Poin=1
23020 Sg1: IF NOT FNExtract(A$,Main$,Poin) THEN Sg2
23040      Out(A$,@Final,"F")
23060      GOTO Sg1
23080 Sg2: IF NOT Sbytes THEN Yasta
23100      Poin=1
23120 Sg3: IF NOT FNExtract(A$,Sub$,Poin) THEN Yasta
23140      Out(A$,@Final,"F")
23160      GOTO Sg3
23180 Yasta: ASSIGN @Final TO *
23200        PURGE "W"&New$
23220        PURGE "S"&New$
23240 SUBEND
23260 !-------------------------------------------------------------------------
23280 ! Append añade T$ a U$, a partir de P
23300 !-------------------------------------------------------------------------
23320 SUB Append(T$,U$,P)
23340      L=LEN(T$)+1
23360      U$[P;L]=CHR$(L-1)&T$
23380      P=P+L
23400 SUBEND
23420 !-------------------------------------------------------------------------
23440 ! FNExtract extrae la string T$ de U$ a partir de P, vuelve con la
23460 !            longitud de T$, y actualiza el puntero
23480 !-------------------------------------------------------------------------
23500 DEF FNExtract(T$,U$,P)
23520      L=NUM(U$[P])
23540      IF NOT L THEN RETURN L
23560      T$=U$[P+1;L]
23580      DISP T$
23600      P=P+L+1
23620      RETURN L
23640 FNEND
23660 !
```